Міністерство освіти і науки України
Національний університет водного господарства
та природокористування
Навчально-науковий механічний інститут
Кафедра розробки родовищ та видобування корисних копалин

02-06-46

МЕТОДИЧНІ ВКАЗІВКИ

до лабораторних робіт із навчальної дисципліни
«Інформаційні технології в гірництві»
для здобувачів вищої освіти другого (магістерського) рівня за спеціальністю 184 «Гірництво»
денної та заочної форми навчання.
Частина 1

Рекомендовано
методичною комісією
зі спеціальності
184 «Гірництво»
Протокол № 8
від 20.02.2019 р.

Рівне – 2019

Методичні вказівки до лабораторних робіт із навчальної дисципліни «Інформаційні технології в гірництві» для здобувачів вищої освіти другого (магістерського) рівня за спеціальністю 184 «Гірництво» денної та заочної форми навчання. В 2 частинах. Частина 1 / Заєць В. В., Семенюк В. В., Оксенюк Р. Р. – Рівне : НУВГП, 2019. – 29 с.

Укладачі:

Заєць В. В., к.т.н., доцент кафедри розробки родовищ та видобування корисних копалин;

Семенюк В. В., асистент кафедри розробки родовищ та видобування корисних копалин;

Оксенюк Р. Р., асистент кафедри розробки родовищ та видобування корисних копалин.

Відповідальний за випуск: В. Я. Корнієнко, професор, д.т.н., в.о. завідувача кафедри розробки родовищ та видобування корисних копалин.

Зміст

Вступ

Велика кількість інженерних та наукових задач вимагає розв'язку лінійних та нелінійних рівнянь та систем рівнянь, наближеного обчислення та інтегрування функцій, розв'язку диференціальних рівнянь, відшукання екстремумів функції з певними обмеженнями, знаходження оптимальних рішень тощо. Для розв'язку таких задач успішно застосовуються засоби обчислювальної техніки.

Сьогодні широко використовуються методи реалізації розв'язування перерахованих задач з використанням комп'ютерної техніки. Такі методи отримали назву чисельних методів. Як правило, розв'язування задач з використанням чисельних методів здійснюється або на базі алгоритмічних мов програмування, або з використанням спеціалізованих пакетів прикладних програм (Mathlab, Mathcad тощо).

Поява такого потужного засобу інженерних та наукових обчислень, як табличний процесор (ТП) Microsoft Excel дозволяє реалізовувати чисельні методи з його допомогою.

В процесі виконання представлених лабораторних робіт студенти матимуть змогу познайомитись з чисельними методами та їх реалізацією в Microsoft Excel на прикладах розв'язку рівнянь, систем рівнянь, інтегрування функцій, інтерполяції та екстраполяції, а також на прикладах розв'язування деяких класів оптимізаційних задач.

У підготовці методичних вказівок використані науково-методичні праці Невзорова А.В., Марченка В.П., Журило С.В.

Лабораторна робота № 1
Інтерполяція функцій із рівновіддаленими вузлами

Мета роботи: Набути практичних навичок при здійсненні інтерполяції функцій, заданих табличним способом, з використанням табличного процесора MS Excel.

Приклад виконання роботи.

Задано табличним способом функцію y=f(x) з рівновіддаленими вузлами x_i (табл. 1.1).

Таблиця 1.1.

k	0	1	2	3	4	5	6	7	8	9	10
x_k	0	0,5	1	1,5	2	2,5	3	3,5	4	4,5	5
y_k	1,26	2,18	3,05	3,94	4,25	5,12	6,05	6,98	7,84	8,61	9,50

Потрібно розрахувати значення функції y= (x) у проміжній точці x=3,77, тобто знайти f(3,77).

Розв'язок. Для отримання результату скористаємось інтерполяційним поліномом Ньютона, який записується у такому вигляді:

$$P_n(x) = y_0 + \sum_{k=1}^{n} (x - x_0)\cdot(x - x_1)\cdot\ldots\cdot(x - x_{k-1})\cdot\frac{\Delta^k y_0}{k!\cdot h^k} \qquad (1.1)$$

де $\Delta^k y_0$ - кінцева різниця k-го порядку:

$$\begin{aligned}
&\Delta^1 y_i = y_{i+1} - y_i; \qquad && i = \overline{0; 1; \ldots; n-1};\\
&\Delta^2 y_j = \Delta^1 y_{j+1} - \Delta^1 y_j; && j = \overline{0; 1; \ldots; n-2};\\
&\ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots\ldots;\\
&\Delta^k y_m = \Delta^{k-1} y_{m+1} - \Delta^{k-1} y_m;\ && m = \overline{0; 1; \ldots; n-m}.
\end{aligned} \qquad (1.2)$$

Для нашої задачі n=10, відповідно, інтерполяційний поліном буде мати 10-й порядок. У зв'язку з цим, для спрощення процесу обчислень доцільно використати засоби обчислювальної техніки, а саме ТП MS Excel.

Створимо на робочому аркуші таблицю та заповнимо її даними, як показано на рис. 1.1.

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1								<u>ІНТЕРПОЛЯЦІЯ</u>							
2					*Крок інтерполяції, h=*	**0,5**	*Поточне значення х:*			**3,77**					
3	**№ з/п**	**Вихідні дані**		*КІНЦЕВІ РІЗНИЦІ*										**Проміжні коеф-ти**	**Коеф-ти полінома Ньютона**
4		**x**	**y**	**1 пор.**	**2 пор.**	**3 пор.**	**4 пор.**	**5 пор.**	**6 пор.**	**7 пор.**	**8 пор.**	**9 пор.**	**10 пор.**		
5	0	0	1,26	0,92	-0,05	0,07	-0,67	2,41	-5,79	11,25	-19,2	30,27	-44,73	7,54	6,9368
6	1	0,5	2,18	0,87	0,02	-0,6	1,74	-3,38	5,46	-7,99	11,03	-14,46		24,6558	-1,23279
7	2	1	3,05	0,89	-0,58	1,14	-1,64	2,08	-2,53	3,04	-3,43			45,53104	3,18717308
8	3	1,5	3,94	0,31	0,56	-0,5	0,44	-0,45	0,51	-0,39				51,67773	-34,624082
9	4	2	4,25	0,87	0,06	-0,06	-0,01	0,06	0,12					36,58784	88,1766856
10	5	2,5	5,12	0,93	9E-16	-0,07	0,05	0,18						15,48885	-89,680446
11	6	3	6,05	0,93	-0,07	-0,02	0,23							3,407547	38,3349055
12	7	3,5	6,98	0,86	-0,09	0,21								0,230009	-4,4253815
13	8	4	7,84	0,77	0,12									-0,011756	-0,3558553
14	9	4,5	8,61	0,89										0,001716	-0,0767737
15	10	5	9,5												
16															
17														**Сума коефіцієнтів полінома Ньютона:**	6,24023562
18															
19														**Значення полінома Ньютона:**	7,50023562
20															
21															

Рис. 1.1. Робочий аркуш з результатами обчислення полінома Ньютона

Розглянемо порядок введення вихідних даних та формул для обчислення проміжних та кінцевих результатів до робочого аркуша MS Excel.

1. Об’єднаємо комірки **A1:O1** і введемо коментар **«ІНТЕРПОЛЯЦІЯ»**.

2. Об’єднаємо комірки **A2:E2** і введемо коментар **«Крок інтерполяції»**. До комірки **F2** введемо значення кроку, яке для наведеного прикладу дорівнює **0,5**. Об’єднаємо комірки **G2:I2**, введемо коментар **«Поточне значення х»**. До комірки **J2** введемо поточне значення, виходячи з умов задачі (для наведеного прикладу воно рівне **3,77**).

3. До діапазону комірок **B5:C15** введемо вихідні дані (значення аргументу та функції у вузлах інтерполяції).

4. Оскільки для наведеного прикладу відомо 10 вузлових точок, потрібно обрахувати 10 кінцевих різниць. Обрахунок їх здійснимо у діапазоні комірок **D5:M14**. Для цього у комірку **D5** введемо формулу **=C6-C5**, після чого скопіюємо її у решту комірок, які повинні містити кінцеві різниці, як показано на рис. 1.1.

5. Введемо формули для обрахунку проміжних коефіцієнтів. З виразу (1.1), який описує поліном Ньютона можна виділити загальну закономірність обчислення коефіцієнтів. Кожний наступний коефіцієнт K_{i+1}-й відрізняється від попереднього K-го множенням на його множник ($x-x_i$) і діленням на коефіцієнт $i{\cdot}h$ (нагадаємо, що $k!=1{\cdot}2{\cdot}3\ \ldots{\cdot}k$, $0!=1,\ 1!=1,\ 2!=1\cdot 2=2,\ldots$ і т.д.).

5.1. Введення формули для 1-го проміжного коефіцієнта (комірка **N5**).

а) У комірку **N5** вводимо формулу для обрахунку першого проміжного коефіцієнта $\left(\dfrac{x-x_0}{1!{\cdot}h^1}\right)$. Для цього у комірку **N5** запишемо формулу **=($J$2–B5)/(A5+1)/$F$2**.

У комірці **J2** знаходиться поточне значення х. При копіюванні адреса цієї комірки повинна залишатись незмінною, тому при її введенні у якості аргументу до формули потрібно використати абсолютну адресацію (після введення адреси комірки натиснути клавішу **F4** на клавіатурі). Те ж саме стосується комірки **F2**, у якій записано крок інтерполяції (різницю між двома сусідніми вузловими точками).

Оскільки 1 !=1, то для того, щоб задати цю константу, використаємо порядковий номер вихідних даних, збільшений на одиницю: **(A5+1)=0+1=1=1!**.

5.2. Введення формули для 2-го проміжного коефіцієнта(комірка **N6**).

У комірку **N6** вводимо формулу для обчислення наступного проміжного коефіцієнта:

$$\frac{(x-x_0)\cdot(x-x_1)}{2!\cdot h^2}=\frac{(x-x_0)}{1\cdot h}\cdot\frac{(x-x_1)}{2\cdot h}=a\cdot b, \qquad (1.3)$$

де **a** - коефіцієнт у комірці **N5**;

b - коефіцієнт, на який потрібно домножити **N5**, тобто

$$b=\frac{(x-x_1)}{2\cdot h}.$$

Тому у комірку **N6** вводимо формулу виду **=N5*($J$2–B6)/(A6+1)/$F$2**.

5.3. Введення решти проміжних коефіцієнтів.

а) Копіюємо формулу з комірки **N6** у решту комірок, що повинні містити проміжні коефіцієнти (у нашому випадку це діапазон **N7:N14**).

б) Якщо задана функція містить кількість вузлових точок, відмінну від 10, відповідно потрібно скорегувати кількість комірок, до яких вводяться формули для обрахунку проміжних коефіцієнтів.

6. Введення формул для обчислення поліному Ньютона.

а) Перший член поліному Ньютона (комірка **O5**) дорівнює

$$(x-x_0)\cdot\frac{\Delta^1 y_0}{1!\cdot h},$$

тобто вміст комірки **N5** потрібно помножити на вміст комірки **D5**, у якій записано кінцеву різницю першого порядку $\Delta^1 y_0$. Для цього в **O5** введемо формулу **=N5*D$5**. Знак $ перед номером рядка означає, що при копіюванні цієї формули до інших комірок діапазону з коефіцієнтами полінома Ньютона номер рядка не повинен змінюватись (потрібно двічі натиснути клавішу **F4** на клавіатурі після введення цього аргументу).

б) Введення формул для обрахунку решти членів поліному Ньютона. Копіюємо формулу з **O5** до комірок **O6:O14**. При цьому комірка **O6** повинна містити формулу =**N6*E$5**, а в ній записано **=N6*D$5**. Комірка **O7** відповідно повинна містити формулу **=N7*F7**, і т.д. Тому після копіювання почергово у кожній комірці діапазону **O6:O14** корегуємо отримані формули, міняючи адресу другого множника (переключаємось на англійську розкладку клавіатури, і у рядку формул змінюємо літеру **D** на потрібну).

7. У комірці **O17** знайдемо суму коефіцієнтів полінома Ньютона, використавши функцію **=СУММ(O5:O14)**.

8. У комірці **O19** розрахуємо значення полінома Ньютона. Для цього до суми його коефіцієнтів (комірка **O17**) додамо y_0 (комірка **C5**): **=O17+C5.**

Таким чином, після введення усіх вихідних даних та формул для обчислення полінома Ньютона, отримаємо результат для поточного значення x: f(3,77)=7,500236. Змінюючи поточне значення x у комірці **J2**, можна отримати значення полінома Ньютона для інших поточних значень аргументу, відмінних від значень у ключових точках.

Для того, щоб впевнитись у правильності виконання обчислень, достатньо у комірку **J2** підставити значення аргументу x_i у вузлових точках. В цьому випадку ми повинні отримати відповідні ним значення y_i, отримані за результатами експерименту або задані умовою задачі.

Завдання для самостійного виконання.

1. Створіть книгу MS Excel, перейменуйте її, надавши їй ім’я ЛР_1_Прізвище (наприклад, ЛР_1_Семеренко).

2. На першому робочому аркуші створіть таблицю і заповніть її даними для задачі із методичних рекомендацій (табл. 1.1). Розрахуйте інтерпольоване значення функції у поточній точці x=3,77, використовуючи поліном Ньютона (використайте формули, як показано на рис. 1.2).

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O
1									**<u>ІНТЕРПОЛЯЦІЯ</u>**						
2					***Крок інтерполяції, h=***	**0,5**		***Поточне значення x:***		**3,77**					
3	**№ з/п**	**Вихідні дані**						***КІНЦЕВІ РІЗНИЦІ***						**Проміжні коеф-ти**	**Коеф-ти полінома Ньютона**
4		**x**	**y**	**1 пор.**	**2 пор.**	**3 пор.**	**4 пор.**	**5 пор.**	**6 пор.**	**7 пор.**	**8 пор.**	**9 пор.**	**10 пор.**		
5	0	0	1,26	=C6-C5	=D6-D5	=E6-E5	=F6-F5	=G6-G5	=H6-H5	=I6-I5	=J6-J5	=K6-K5	=L6-L5	=($J$2-B5)/(A5+1)/$F$2	=N5*D$5
6	1	0,5	2,18	=C7-C6	=D7-D6	=E7-E6	=F7-F6	=G7-G6	=H7-H6	=I7-I6	=J7-J6	=K7-K6		=N5*($J$2-B6)/(A6+1)/$F$2	=N6*E$5
7	2	1	3,05	=C8-C7	=D8-D7	=E8-E7	=F8-F7	=G8-G7	=H8-H7	=I8-I7	=J8-J7			=N6*($J$2-B7)/(A7+1)/$F$2	=N7*F$5
8	3	1,5	3,94	=C9-C8	=D9-D8	=E9-E8	=F9-F8	=G9-G8	=H9-H8	=I9-I8				=N7*($J$2-B8)/(A8+1)/$F$2	=N8*G$5
9	4	2	4,25	=C10-C9	=D10-D9	=E10-E9	=F10-F9	=G10-G9	=H10-H9					=N8*($J$2-B9)/(A9+1)/$F$2	=N9*H$5
10	5	2,5	5,12	=C11-C10	=D11-D10	=E11-E10	=F11-F10	=G11-G10						=N9*($J$2-B10)/(A10+1)/$F$2	=N10*I$5
11	6	3	6,05	=C12-C11	=D12-D11	=E12-E11	=F12-F11							=N10*($J$2-B11)/(A11+1)/$F$2	=N11*J$5
12	7	3,5	6,98	=C13-C12	=D13-D12	=E13-E12								=N11*($J$2-B12)/(A12+1)/$F$2	=N12*K$5
13	8	4	7,84	=C14-C13	=D14-D13									=N12*($J$2-B13)/(A13+1)/$F$2	=N13*L$5
14	9	4,5	8,61	=C15-C14										=N13*($J$2-B14)/(A14+1)/$F$2	=N14*M$5
15	10	5	9,5												
16															
17														**Сума коефіцієнтів полінома Ньютона:**	=СУММ(O5:O15)
18															
19														**Значення полінома Ньютона:**	=O17+C5
20															
21															
22															

Рис. 1.2. Робочий аркуш із введеними формулами для обчислення полінома Ньютона для поточного значення x

3. Скопіюйте перший робочий аркуш у цю ж книгу. Введіть на копії аркуша вихідні дані відповідно до індивідуального завдання (табл. 1.2) і розрахуйте значення полінома Ньютона для поточних проміжних точок відповідно до свого завдання.

4. Підготуйте відповіді на контрольні запитання.

Таблиця 1.2

Варіанти індивідуального завдання

k	0	1	2	3	4	5	6	7	8	9	10
Варіант 1											
x_k	1,410	1,415	1,420	1,425	1,430	1,435	1,440	1,445	1,450	1,455	1,460
y_k	0,870	0,880	0,889	0,890	0,891	0,892	0,893	0,894	0,895	0,896	0,897
Поточне значення x					X_1=1,412			X_2=1,437			X_3=1,453
Варіант 2											
x_k	0,096	0,101	0,106	0,111	0,116	0,121	0,126	0,131	0,136	0,141	0,146
y_k	1,252	1,261	1,276	1,291	1,306	1,321	1,336	1,352	1,367	1,383	1,399
Поточне значення x					X_1=1,105			X_2=1,123			X_3=1,144
Варіант 3											
x_k	0,100	0,150	0,200	0,250	0,300	0,350	0,400	0,450	0,500	0,550	0,600
y_k	0,954	0,860	0,819	0,779	0,741	0,705	0,670	0,638	0,606	0,577	0,549
Поточне значення x					X_1=0,154			X_2=0,407			X_3=0,552
Варіант 4											
x_k	0,175	0,180	0,185	0,190	0,195	0,200	0,205	0,210	0,215	0,220	0,225
y_k	5,804	5,615	5,467	5,352	5,193	5,066	4,946	4,832	4,722	4,618	4,519
Поточне значення x					X_1=0,182			X_2=0,203			X_3=0,224

k	0	1	2	3	4	5	6	7	8	9	10
Варіант 5											
x_k	3,450	3,500	3,550	3,600	3,650	3,700	3,750	3,800	3,850	3,900	3,950
y_k	31,850	33,110	34,650	36,600	38,470	40,440	42,520	44,700	46,990	49,400	51,930
Поточне значення x					X_1=3,506			X_2=3,752			X_3=3,903
Варіант 6											
x_k	0,110	0,115	0,120	0,125	0,130	0,135	0,140	0,145	0,150	0,155	0,160
y_k	8,810	8,680	8,290	7,960	7,650	7,360	7,100	6,850	6,620	6,400	6,180
Поточне значення x					X_1=0,122			X_2=0,144			X_3=0,157
Варіант 7											
x_k	1,335	1,340	1,345	1,350	1,355	1,360	1,365	1,370	1,375	1,380	1,385
y_k	4,120	4,260	4,350	4,460	4,560	4,670	4,790	4,910	5,020	5,180	5,310
Поточне значення x					X_1=1,347			X_2=1,362			X_3=1,381
Варіант 8											
x_k	0,140	0,150	0,160	0,170	0,180	0,190	0,200	0,210	0,220	0,230	0,240
y_k	3,970	4,480	4,950	5,470	5,990	6,050	6,680	6,909	7,380	8,166	9,025
Поточне значення x					X_1=0,164			X_2=0,203			X_3=0,235
Варіант 9											
x_k	0,440	0,450	0,460	0,470	0,480	0,490	0,500	0,510	0,520	0,530	0,540
y_k	20,730	20,190	19,610	18,940	18,170	17,300	16,310	15,190	13,940	12,550	10,920
Поточне значення x					X_1=0,477			X_2=0,494			X_3=0,528
Варіант 10											
x_k	0,010	0,060	0,110	0,160	0,210	0,260	0,310	0,360	0,410	0,460	0,510
y_k	0,990	0,950	0,910	0,880	0,840	0,810	0,780	0,740	0,710	0,680	0,590
Поточне значення x					X_1=0,163			X_2=0,334			X_3=0,475

Продовження таблиці 1.2

Варіант 11											
x_k	0,122	0,124	0,126	0,128	0,130	0,132	0,134	0,136	0,138	0,140	0,142
y_k	3,120	3,320	3,560	3,870	4,060	4,390	4,770	5,120	5,440	5,810	6,350
Поточне значення x			$X_1=0,125$			$X_2=0,135$			$X_3=0,139$		
Варіант 12											
x_k	5,270	5,320	5,370	5,420	5,470	5,520	5,570	5,620	5,670	5,720	5,770
y_k	4,440	4,560	4,650	4,770	4,860	4,990	5,110	5,200	5,310	5,440	5,590
Поточне значення x			$X_1=5,388$			$X_2=5,585$			$X_3=5,746$		
Варіант 13											
x_k	2,58	2,61	2,64	2,67	2,7	2,73	2,76	2,79	2,82	2,85	2,88
y_k	1,440	1,510	1,580	1,630	1,710	1,790	1,900	2,010	2,110	2,220	2,350
Поточне значення x			$X_1=2,652$			$X_2=2,745$			$X_3=2,833$		
Варіант 14											
x_k	6,17	6,21	6,25	6,29	6,33	6,37	6,41	6,45	6,49	6,53	6,57
y_k	4,09	4,17	4,26	4,31	4,37	4,48	4,54	4,65	4,72	4,8	4,92
Поточне значення x			$X_1=6,274$			$X_2=6,384$			$X_3=6,544$		
Варіант 15											
x_k	6,21	6,27	6,33	6,39	6,45	6,51	6,57	6,63	6,69	6,75	6,81
y_k	8,18	8,27	8,39	8,4	8,49	8,53	8,64	8,71	8,85	8,97	9,05
Поточне значення x			$X_1=6,41$			$X_2=6,62$			$X_3=6,7$		
Варіант 16											
x_k	7,54	7,58	7,62	7,66	7,7	7,74	7,78	7,82	7,86	7,9	7,94
y_k	12,5	12,8	13	13,4	13,9	14,5	15,1	15,4	15,9	16,5	17,2
Поточне значення x			$X_1=7,61$			$X_2=7,75$			$X_3=7,88$		

Контрольні запитання.

1. Що таке вузлові точки для інтерполяційної задачі?
2. Що таке критерій найменших квадратів? Для яких випадків інтерполяції він використовується?
3. Як формулюється задача інтерполяції?
4. Що таке кінцеві різниці? Який найбільший порядок може мати кінцева різниця у задачах інтерполяції?
5. Запишіть формулу інтерполяційного полінома Ньютона для рівновіддалених вузлів. Поясніть, які елементи входять до неї.
6. Яким виразом оцінюється похибка при заміні функції її інтерполяційним поліномом?
7. Який порядок має інтерполяційний поліном Ньютона?
8. Зобразіть графічну інтерпретацію розв'язування задачі інтерполяції. Поясніть зображення на графіку.

Лабораторна робота № 2

Екстраполяція функцій із рівновіддаленими вузлами

Мета: Набути практичних навичок при здійсненні екстраполяції функцій, заданих табличним способом, з використанням табличного процесора MS Excel.

Приклад виконання роботи.

Задано табличним способом функцію y=f(x) з рівновіддаленими вузлами x_i (табл. 2.1).

Таблиця 2.1

Табличні значення функції y = f (x) з рівновіддаленими вузлами

k	**0**	**1**	**2**	**3**	**4**	**5**
x_k	0	2	4	6	8	10
y_k	7	12	15	18	20	21

Розрахувати прогнозні значення функції y=f(x) на два періоди вперед, тобто для x_6=12 та x_7=14.

Розв'язок. Для прогнозування (екстраполяції) функції, заданої у табличній формі, у ТП MS Excel передбачено ряд можливостей.

Розглянемо спосіб за допомогою вбудованих функцій.

Для прогнозування (екстраполяції) даних у ТП MS Excel використовуються функції ТЕНДЕНЦИЯ() і ПРЕДСКАЗ().

Відмінність між цими функціями полягає в тому, що функція ТЕНДЕНЦИЯ() дозволяє отримати результат екстраполяції на довільне вказане число періодів, а функція ПРЕДСКАЗ() − лише для одного прогнозного значення аргументу xj (j>n, де n − кількість вузлових точок заданої функціональної залежності)

Здійснимо прогноз функції, заданої у табл. 2.1, за допомогою цих функцій. Для цього насамперед на робочому аркуші створимо таблицю із вихідними даними і екстраполяційні точки.

У створену таблицю у комірки для екстрапольованих значень функції f(x) введемо спочатку функцію ТЕНДЕНЦИЯ(). Порядок введення аргументів, таблицю із введеною функцією і таблицю з отриманими результатами представлено на рис. 2.1.



а)



б)



в)

Рис. 2.1. Зразок використання функції ТЕНДЕНЦИЯ(): а) введення аргументів у вікно майстра; б) вигляд аркуша із введеною функцією; в) вигляд аркуша із результатами прогнозу

Оскільки функція ТЕНДЕНЦИЯ() є функцією масиву, порядок її використання такий:

1) виділимо комірки, де мають бути отримані результати екстраполяції (для даного прикладу – комірки **H5:I5**);

2) викликаємо відомим способом майстер функцій і вводимо аргументи, як показано на рис. 2.1, а);

3) натискаємо комбінацію клавіш Ctrl+Shift+Enter. Для закінчення роботи з майстром і отримання результату прогнозу. Результат показано на рис. 2.1, в).

Розглянемо порядок екстраполяції з використанням функції ПРЕДСКАЗ(). Як вказувалось вище, ця функція дозволяє отримувати лише одне екстрапольоване значення. Тому для отримання прогнозу на два періоди її потрібно використати окремо для комірки **H5** та **I5**.

Порядок введення аргументів функції ПРЕДСКАЗ(), таблицю із введеною функцією та таблицю з отриманими результатами показано на рис. 2.2.



а)

H5 fx =ПРЕДСКАЗ(H4;$B$5:$G$5;$B$4:$G$4)

	C	D	E	F	G	H	I
1	ЕКСТРАПОЛЯЦІЯ (ПРОГНОЗ)						
2	Вихідні дані					Прогнозні дані	
3	1	2	3	4	5	6	7
4	2	4	6	8	10	12	14
5	12	15	18	20	21	=ПРЕДСКАЗ(H4;$B$5:$G$5;$B$4:$G$4)	=ПРЕДСКАЗ(I4;$B$5:$G$5;$B$4:$G$4)

б)

	A	B	C	D	E	F	G	H	I
1	ЕКСТРАПОЛЯЦІЯ (ПРОГНОЗ)								
2		Вихідні дані						Прогнозні дані	
3	k	0	1	2	3	4	5	6	7
4	Xk	0	2	4	6	8	10	12	14
5	Yk	7	12	15	18	20	21	25,2	27,97143

в)

Рис. 2.2. Зразок використання функції ПРЕДСКАЗ( ): а) введення аргументів у вікно майстра; б) вигляд аркуша із введеною функцією; в) вигляд аркуша із результатами прогнозу

Для використання функції ПРЕДСКАЗ( ) виконуємо таку послідовність дій:

і) встановлюємо курсор у першу комірку, у якій потрібно отримати екстрапольоване значення (для нашого прикладу це комірка **H5**);

2) за допомогою майстра функцій викликаємо вікно функції ПРЕДСКАЗ() і вводимо аргументи, як показано на рис. 2.2, а);

3) натискаємо кнопку ОК або клавішу Enter на клавіатурі;

4) копіюємо за допомогою режиму автозаповнення (або через буфер обміну) вміст комірки **H5** до комірки **I5**. Отримуємо результат, як показано на рис. 2.2, в).

Порівнюючи результати прогнозу, отримані за допомогою функцій ТЕНДЕНЦИЯ() і ПРЕДСКАЗ(), можна впевнитись у тому, що вони повністю співпадають. Тому жорстких вимог щодо того, яку з вказаних функцій слід використовувати, не існує – все залежить від конкретної задачі.

Завдання для самостійного виконання.

1. Створіть книгу MS Excel, перейменуйте її, надавши їй ім'я ЛР_2_Прізвище (наприклад, ЛР_2_Семеренко).
2. На першому робочому аркуші створіть таблицю і заповніть її даними для задачі із методичних рекомендацій (табл. 2.2). Розрахуйте екстрапольовані значення функції на 2 періоди, використовуючи функцію ТЕНДЕНЦИЯ().
3. Скопіюйте перший робочий аркуш у цю ж книгу. Виконайте аналогічно екстраполяцію з використанням функції ПРЕДСКАЗ().
4. На окремому робочому аркуші створіть дві аналогічні таблиці відповідно до індивідуального варіанту (табл. 2.2) та проведіть екстраполяцію у цих таблицях відповідно з використанням функцій ТЕНДЕНЦИЯ() і ПРЕДСКАЗ().
5. Підготуйте відповіді на контрольні запитання.

Контрольні запитання

1. Особливості використання функції ТЕНДЕНЦИЯ().
2. Особливості використання функції ПРЕДСКАЗ().
3. Поняття екстраполяції.

4. Поняття лінії тренду, порядок її побудови.

Таблиця 2.2.

Варіанти індивідуального завдання

k	**0**	**1**	**2**	**3**	**4**	**5**	***k***	**0**	**1**	**2**	**3**	**4**	**5**
Варіант 1							***Варіант 9***						
x_k	0	2	4	6	8	10	x_k	10	15	20	25	30	35
y_k	12,5	13,6	14,8	15,4	16,5	17,6	y_k	18,6	20,3	21,9	23,6	25,2	27,1
Варіант 2							***Варіант 10***						
x_k	2	6	10	14	18	22	x_k	13	16	19	22	25	28
y_k	7,8	7,4	7,2	6,9	6,4	6,2	y_k	29,5	28,3	27,4	26,4	25,2	24,4
Варіант 3							***Варіант 11***						
x_k	4	7	10	13	16	19	x_k	12	16	20	24	28	32
y_k	18,5	17,9	17,5	16,9	16,3	15,9	y_k	32,5	32,9	33,7	34,6	35,3	36,1
Варіант 4							***Варіант 12***						
x_k	5	8	11	14	17	20	x_k	8	12	16	20	24	28
y_k	6,44	6,53	6,62	6,65	6,72	6,81	y_k	1,86	1,95	1,99	2,05	2,11	2,19
Варіант 5							***Варіант 13***						
x_k	3	6	9	12	15	18	x_k	11	16	21	26	31	36
y_k	3,4	3,55	3,66	3,81	3,99	4,11	y_k	14,8	14,6	14,2	13,8	13,7	13,4
Варіант 6							***Варіант 14***						
x_k	1	5	9	13	17	21	x_k	10	13	16	19	22	25
y_k	8,45	8,54	8,66	8,72	8,81	8,86	y_k	10	10,6	11	11,3	11,9	12,2
Варіант 7							***Варіант 15***						
x_k	3	7	11	15	19	23	x_k	6	9	12	15	18	21
y_k	14,6	14,1	13,8	13,7	13,1	12,2	y_k	11,1	12,2	13	13,5	15,1	16,2
Варіант 8							***Варіант 16***						
x_k	1	6	11	16	21	26	x_k	9	15	21	27	33	39
y_k	9,6	10,8	11,3	11,9	12,9	13,4	y_k	21,2	21,9	22,6	23,3	24,2	24,5

Лабораторна робота № 3
Наближений розв'язок нелінійних рівнянь

Мета: Набути практичних навичок при знаходженні коренів нелінійних рівнянь чисельними методами з використанням табличного процесора MS Excel.

Приклад виконання роботи.

Задано нелінійне рівняння з однією змінною виду f(x)=0. Уточнити корінь цього рівняння на інтервалі [a; b], якщо відомо, що на цьому інтервалі існує лише один корінь:

$$e^{-x} - x^2 = 0; \qquad a{=}0{,}4; \qquad b{=}1{,}2. \tag{3.1}$$

Корінь рівняння уточнити з точністю є=0,0001

Розв'язок. Дану задачу розв'яжемо з використанням надбудови MS Excel «Поиск решения»;

Створимо на робочому аркуші книги MS Excel таблицю, як показано на рис. 3.1

	A	B
1	Уточнення коренів	
2	Режим "Поиск решения"	
3	Початкове наближення кореня	0,4
4	Рівняння	=EXP(-B3)-B3^2
5	Нижня границя інтервалу	0,4
6	Верхня границя інтервалу	1,2
7		

Рис. 3.1. Фрагмент робочого аркуша з вихідними даними для розв'язку рівняння за допомогою надбудови «Поиск решения»

Після створення таблиці виконаємо таку послідовність дій:

1. Встановимо курсор в комірку **B4** і виконаємо команду головного меню **Сервисы-Поиск решения**. В результаті з'явиться діалогове вікно, у якому встановимо параметри, як показано на рис. 3.2.

2. У діалоговому вікні «**Поиск решения**» натиснемо кнопку **Праралетры** і встановимо налаштування, як показано на рис. 3.3.

3. У вікні «**Параметры поиска решения**» натиснемо кнопку **ОК**.

4. У вікні «**Поиск решения**» натиснемо кнопку **Выполнить**



Рис. 3.2. Діалогове вікно «Поиск решения» із встановленими вихідними даними для розв'язку рівняння



Рис. 3.3. Налаштування режиму «Поиск решения»

В результаті виконаних дій, у комірці **B4** отримаємо чисельний розв'язок рівняння на інтервалі [0,4; 1,2] (рис. 3.4).

	A	B
1	**Уточнення коренів**	
2	**Режим "Поиск решения"**	
3	Початкове наближення кореня	0,703467
4	Рівняння	4,12E-07
5	Нижня границя інтервалу	0,4
6	Верхня границя інтервалу	1,2

Рис 3.4. Фрагмент робочого аркуша з результатами розв'язку рівняння методом «Поиск решения»

Як видно з рис. 3.4, на інтервалі [0,4; 1,2] корінь рівняння x=0,703467.

Завдання для самостійного виконання.

1. Створіть книгу MS Excel, перейменуйте її, надавши їй ім'я ЛР_3_Прізвище (наприклад, ЛР_3_Семеренко).

2. На першому робочому аркуші створити відповідну таблицю і уточнити корінь рівняння з індивідуального варіанту (табл. 3.1) з використанням надбудови «**Поиск решения**».

3. Підготувати відповіді на контрольні запитання.

Контрольні запитання.

1. Що таке ітераційний процес?

2. Як визначається похибка при розв'язуванні рівняння методом ітерацій?

3. Графічна інтерпретація збіжності ітераційного процесу.

4. Графічна інтерпретація розбіжності ітераційного процесу.

5. Критерій збіжності ітераційного процесу.

6. Налаштування MS Excel для виконання ітераційних обчислень.

7. Що таке циклічне посилання в MS Excel?

Таблиця 3.1

Завдання для індивідуального виконання

Варіант 1	***Варіант 9***
$\cos^2 x - x^2 = 0$; **a**=0,1 **b**=1,5 ε=0,0001	$y = \cos(\ln x)$; **a**=4,5 **b**=5,5 ε=0,0001
Варіант 2	***Варіант 10***
$\ln x - 1/x = 0$; **a**=1,2 **b**=2,0 ε=0,0001	$y = \sin(\ln x)$; **a**=22,7 **b**=23,5 ε=0,0001
Варіант 3	***Варіант 11***
$y = \ln x + x^2$; **a**=0,3 **b**=0,9 ε=0,0001	$y = \sin(e^x)$; **a**=0,7 **b**=1,4 ε=0,0001
Варіант 4	***Варіант 12***
$y = e^x - x^2$; **a**=0,7 **b**=1,6 ε=0,0001	$y = \cos(e^x)$; **a**=1,0 **b**=2,0 ε=0,0001
Варіант 5	***Варіант 13***
$y = 5\sin x - x$; **a**=2,2 **b**=3,2 ε=0,0001	$y = \ln(x^2 - x)$; **a**=1,3 **b**=1,9 ε=0,0001
Варіант 6	***Варіант 14***
$y = \ln x - \sin x$; **a**=1,8 **b**=2,5 ε=0,0001	$y = 1/x - \ln x$; **a**=1,4 **b**=2,0 ε=0,0001
Варіант 7	***Варіант 15***
$y = 2^x - x^3$; **a**=1,1 **b**=1,8 ε=0,0001	$y = \ln x - 1/x^2$; **a**=1,0 **b**=2,0 ε=0,0001
Варіант 8	***Варіант 16***
$y = x^2 - 3^x$; **a**=1,1 **b**=1,8 ε=0,0001	$y = \ln(1/(x^2 - x))$; **a**=1,0 **b**=2,0 ε=0,0001

Лабораторна робота № 4
Розв’язування системи лінійних алгебраїчних рівнянь засобами MS Excel

Мета: Набути практичних навичок при розв’язуванні систем лінійних алгебраїчних рівнянь (ЛАР) з використанням табличного процесора MS Excel.

Порядок виконання роботи

Знайти корені системи ЛАР:

$$\begin{cases} -8\cdot x_1 + x_2 + 2\cdot x_3 = 0; \\ 5\cdot x_1 + 7\cdot x_2 - 3\cdot x_3 = 10; \\ 2\cdot x_1 + x_2 - 2\cdot x_3 = -2. \end{cases} \quad (4.1)$$

Розв’язок. Дане рівняння з використанням табличного процесора MS Excel достатньо легко може бути розв’язане двома методами:

1) методом оберненої матриці;

2) наближеним методом (методом простих ітерацій).

Розглянемо метод оберненої матриці. За методом оберненої матриці систему ЛАР потрібно переписати у матричній формі:

$$[A]\bullet[X] = \bullet[B],$$

де $[A]=\begin{bmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{bmatrix}=\begin{bmatrix} -8 & 1 & 2 \\ 5 & 7 & -3 \\ 2 & 1 & -2 \end{bmatrix}$; $[X]=\begin{bmatrix} x_1 \\ x_2 \\ x_3 \end{bmatrix}$; $[B]=\begin{bmatrix} b_1 \\ b_2 \\ b_3 \end{bmatrix}=\begin{bmatrix} 0 \\ 10 \\ -2 \end{bmatrix}$ (4.2)

Тоді матриця-вектор невідомих [X] може бути обрахована як

$$[X]=[A]^{-1}\bullet[B], \quad (4.3)$$

де $[A]^{-1}$ – матриця, обернена до матриці [A].

Для розрахунку оберненої матриці $[A]^{-1}$ у ТП MS Excel вбудована функція **МОБР(Масив)**, аргумент якої **Масив** - це діапазон комірок, що містить коефіцієнти прямої матриці [A] .

Для розрахунку добутку двох матриць існує функція **МУМНОЖ(Масив_1;Масив_2)**. Аргументи цієї функції - це діапазони комірок, що містять коефіцієнти відповідних матриць.

Обидві вказані функції є функціями масиву, тому порядок їх використання такий:

1. Виділити діапазон комірок, де повинен міститись результат обчислень;
2. Викликати за допомогою майстра діалогове вікно відповідної функції;
3. Ввести до відповідного поля (полів) адресу діапазону комірок, що є аргументом;
4. Після введення аргументів натиснути комбінаціє клавіш **Ctrl+Shift+Enter**.

Створимо у робочому аркуші книги MS Excel таблиці та введемо вихідні дані та функції, як показано на рис. 4.1.

	A	B	C	D	E
1		**Вихідна матриця [A]**			**Вектор [B]**
2	-8	1	2		0
3	5	7	-3		10
4	2	1	-2		-2
5					
6		**Обернена матриця [1/A]**			**Вектор [X]**
7	=МОБР(A2:C4)	=МОБР(A2:C4)	=МОБР(A2:C4)		=МУМНОЖ(A7:C9;E2:E4)
8	=МОБР(A2:C4)	=МОБР(A2:C4)	=МОБР(A2:C4)		=МУМНОЖ(A7:C9;E2:E4)
9	=МОБР(A2:C4)	=МОБР(A2:C4)	=МОБР(A2:C4)		=МУМНОЖ(A7:C9;E2:E4)

Рис. 4.1. Введення вихідних даних та функцій для обчислення коренів системи ЛАР

Результат обчислення коренів системи показано на рис. 4.2.

	A	B	C	D	E
1	**Вихідна матриця [A]**				**Вектор [B]**
2	-8	1	2		0
3	5	7	-3		10
4	2	1	-2		-2
5					
6	**Обернена матриця [1/A]**				**Вектор [X]**
7	-0,149	0,054	-0,22973		1
8	0,0541	0,162	-0,18919		2
9	-0,122	0,135	-0,82432		3

Рис. 4.2. Результат обчислення коренів системи ЛАР

Таким чином, в результаті проведених обчислень визначили такі корені системи ЛАР: $x_1=1$, $x_2=2$, $x_3=3$.

Завдання для самостійного виконання.

1. Створіть книгу MS Excel, перейменуйте її, надавши їй ім’я ЛР_4_Прізвище (наприклад, ЛР_4_Семеренко).
2. На першому робочому аркуші створити відповідну таблицю і розв’язати систему ЛАР з індивідуального варіанту (табл. 4.1) з використанням методу оберненої матриці.
3. Підготувати відповіді на контрольні запитання.

Таблиця 4.1.

Завдання для індивідуального виконання

Варіант 1	***Варіант 9***
$\begin{cases} 9x_1+3x_2+5x_3=7; \\ 4x_1+7x_2+2x_3=12; \\ 4x_1+6x_2-15x_3=6. \end{cases}$	$\begin{cases} 88x_1+51x_2-22x_3=18; \\ 14x_1+56x_2-25x_3=11; \\ 7x_1+19x_2-29x_3=-9. \end{cases}$
Варіант 2	***Варіант 10***
$\begin{cases} 15x_1+5x_2+2x_3=11; \\ -5x_1+12x_2+4x_3=8; \\ 3x_1+4x_2+11x_3=-5. \end{cases}$	$\begin{cases} 12x_1+5x_2+9x_3=88; \\ 9x_1+22x_2-11x_3=46; \\ 8x_1+45x_2+64x_3=120. \end{cases}$
Варіант 3	***Варіант 11***
$\begin{cases} 51x_1+12x_2+14x_3=-6; \\ 4x_1+21x_2+8x_3=15; \\ 4x_1+12x_2+32x_3=17. \end{cases}$	$\begin{cases} 44x_1+1x_2+9x_3=125; \\ 5x_1+25x_2-9x_3=84; \\ 13x_1-12x_2-64x_3=154. \end{cases}$
Варіант 4	***Варіант 12***
$\begin{cases} 9x_1+5x_2+3x_3=4; \\ 15x_1+19x_2+7x_3=-5; \\ 12x_1+4x_2-18x_3=6. \end{cases}$	$\begin{cases} 58x_1+41x_2+11x_3=320; \\ 11x_1+17x_2+9x_3=54; \\ 7x_1-5x_2-16x_3=-49. \end{cases}$
Варіант 5	***Варіант 13***
$\begin{cases} 11x_1+5x_2+7x_3=4; \\ 9x_1+18x_2+6x_3=-8; \\ 5x_1+7x_2+16x_3=8. \end{cases}$	$\begin{cases} 16x_1+8x_2+4x_3=55; \\ 9x_1+12x_2+7x_3=54; \\ 8x_1+12x_2+18x_3=-8. \end{cases}$
Варіант 6	***Варіант 14***
$\begin{cases} 47x_1+21x_2+11x_3=5; \\ 11x_1+19x_2-6x_3=14; \\ -5x_1+4x_2-17x_3=3. \end{cases}$	$\begin{cases} 41x_1+14x_2+25x_3=75; \\ 15x_1+29x_2+18x_3=54; \\ 16x_1+18x_2-32x_3=18. \end{cases}$
Варіант 7	***Варіант 15***
$\begin{cases} 8x_1+5x_2+3x_3=1; \\ 7x_1+12x_2-5x_3=5; \\ 8x_1+4x_2+12x_3=10. \end{cases}$	$\begin{cases} 44x_1+21x_2+18x_3=56; \\ 5x_1+7x_2+6x_3=18; \\ 2x_1+1x_2-12x_3=14. \end{cases}$

Продовження таблиці 4.1

Варіант 8	***Варіант 16***
$\begin{cases} 54x_1 + 21x_2 + 11x_3 = 20; \\ 12x_1 + 47x_2 - 6x_3 = 14; \\ 12x_1 + 8x_2 - 44x_3 = 5. \end{cases}$	$\begin{cases} 65x_1 + 24x_2 - 16x_3 = 7; \\ 4x_1 + 7x_2 + 2x_3 = 12; \\ 4x_1 + 6x_2 - 15x_3 = 6. \end{cases}$

Контрольні запитання

1. Яка матриця називається оберненою? Як вона визначається?
2. Як записується система ЛАР в ітераційній формі?
3. Як обраховується ітераційна матриця?
4. Критерій збіжності для методу простих ітерацій.
5. Метод Зейделя наближеного розв’язку систем ЛАР.

Лабораторна робота № 5

Наближене інтегрування функцій із заданим кроком

Мета: Набути практичних навичок при розрахунку визначених інтегралів чисельними методами з використанням табличного процесора MS Excel.

Приклад виконання роботи

Розрахувати значення визначеного інтегралу методом прямокутників з числом кроків n=10. Порівняти отримані результати.

$$I = \int_0^1 (9 \cdot x^8 + 5 \cdot x^4 + 3 \cdot x^2 + 7) dx \tag{5.1}$$

Істинне значення даного інтегралу дорівнює 10.

Розв’язок. Розглянемо вказаний метод чисельного інтегрування.

Метод прямокутників.

Запишемо визначений інтеграл у загальному вигляді:

$$I = \int_a^b f(x) dx . \tag{5.2}$$

Розіб’ємо інтервал [a; b] на *n* рівних підінтервалів. Тоді ширина кожного підінтервалу буде рівною *h = (b-a)/n*. Точки розбиття матимуть координати

$$x_1 = a;\ x_2 = a + h;\ x_3 = a + 2h; ...;\ x_i = a + (i-1) \cdot h; ...;\ x_{n+1} = a + n \cdot h = b .$$

За методом прямокутників значення визначеного інтегралу розраховується як

$$I = \sum_{i=1}^{n} f(x_i) \cdot h . \tag{5.3}$$

Згідно з цим виразом складемо таблицю у робочому аркуші MS Excel та заповнимо її даними, як показано на рис. 5.1, а).

Результати обрахунків значення визначеного інтегралу за методом прямокутників показано на рис. 5.1, б).

	A	B	C	D	E	F	G
1	$I = \int_0^1 (9 \cdot x^8 + 5 \cdot x^4 + 3 \cdot x^2 + 7)\,dx$				***Нижня границя***		0
2					***Верхня границя***		1
3					***Кількість підінтервалів, n***		10
4					***Ширина підінтервалу, h***		=(G2-G1)/G3
5	***i***	***Xi***	***f (Xi)***	***Площа Si***	***Істинне значення інтегралу***		10
6	1	=$G$1	=9*B6^8+5*B6^4+3*B6^2+7	=C6*$G$4	***Розраховане значення***		=СУММ(D6:D15)
7	2	=B6+$G$4	=9*B7^8+5*B7^4+3*B7^2+7	=C7*$G$4	***Абсолютна похибка***		=ABS(G5-G6)
8	3	=B7+$G$4	=9*B8^8+5*B8^4+3*B8^2+7	=C8*$G$4	***Відносна похибка***		=G7/G5
9	4	=B8+$G$4	=9*B9^8+5*B9^4+3*B9^2+7	=C9*$G$4			
10	5	=B9+$G$4	=9*B10^8+5*B10^4+3*B10^2+7	=C10*$G$4			
11	6	=B10+$G$4	=9*B11^8+5*B11^4+3*B11^2+7	=C11*$G$4			
12	7	=B11+$G$4	=9*B12^8+5*B12^4+3*B12^2+7	=C12*$G$4			
13	8	=B12+$G$4	=9*B13^8+5*B13^4+3*B13^2+7	=C13*$G$4			
14	9	=B13+$G$4	=9*B14^8+5*B14^4+3*B14^2+7	=C14*$G$4			
15	10	=B14+$G$4	=9*B15^8+5*B15^4+3*B15^2+7	=C15*$G$4			
16	11	=B15+$G$4					

а)

	A	B	C	D	E	F	G
1	$I = \int_0^1 (9 \cdot x^8 + 5 \cdot x^4 + 3 \cdot x^2 + 7)\,dx$				***Нижня границя***		0
2					***Верхня границя***		1
3					***Кількість підінтервалів, n***		10
4					***Ширина підінтервалу, h***		0,1
5	***i***	***Xi***	***f (Xi)***	***Площа Si***	***Істинне значення інтегралу***		10
6	1	0	7	0,7	***Розраховане значення***		9,231232
7	2	0,1	7,0305	0,70305	***Абсолютна похибка***		0,768768
8	3	0,2	7,128023	0,7128023	***Відносна похибка***		0,076877
9	4	0,3	7,31109	0,731109			
10	5	0,4	7,613898	0,7613898			
11	6	0,5	8,097656	0,8097656			
12	7	0,6	8,879165	0,8879165			
13	8	0,7	10,18933	1,0189332			
14	9	0,8	12,47795	1,2477949			
15	10	0,9	16,5847	1,6584705			
16	11	1					

б)

Рис. 5.1. Розрахункові формули а) та результати обчислення б) визначеного інтегралу за методом прямокутників

Як видно з рис. 5.1, при кількості інтервалів *n=10* відносна похибка обчислення визначеного інтегралу за методом прямокутників склала 0,0769, або 7,69 %.

Завдання для самостійного виконання.

1. Створіть книгу MS Excel, перейменуйте її, надавши їй ім’я ЛР_5_Прізвище (наприклад, ЛР_5_Семеренко).
2. На першому робочому аркуші створити відповідну таблицю і розрахувати значення визначеного інтегралу відповідно до індивідуального варіанту (табл. 5.1) з використанням методу прямокутників.
3. Підготувати відповіді на контрольні запитання.

Таблиця 5.1

Завдання для індивідуального виконання

Інтеграл та його істинне значення	***n***	**Інтеграл та його істинне значення**	***n***
Варіант 1		***Варіант 9***	
$I=\int\limits_1^3(\frac{1}{x^2}+x)dx$, $I_{\text{іст.}}=4{,}6666667$	10	$I=\int\limits_{0,5}^{2,5}\frac{dx}{\sqrt{2x^2+1}}$, $I_{\text{іст.}}=0{,}9312299$	10
Варіант 2		***Варіант 10***	
$I=\int\limits_1^3 1/(x^2+x)dx$, $I_{\text{іст.}}=0{,}4054651$	10	$I=\int\limits_2^5\frac{x}{\ln x}dx$, $I_{\text{іст.}}=8{,}5453714$	10
Варіант 3		***Варіант 11***	
$I=\int\limits_1^6(\ln x+x^2)dx$, $I_{\text{іст.}}=77{,}4172233$	10	$I=\int\limits_1^6\frac{\cos x+2}{x^2}dx$, $I_{\text{іст.}}=1{,}5683361$	10
Варіант 4		***Варіант 12***	
$I=\int\limits_0^3(\frac{1}{x^3+1}+x)dx$, $I_{\text{іст.}}=5{,}6544498$	10	$I=\int\limits_3^6\frac{\sin^2 x+1}{x^2+1}dx$, $I_{\text{іст.}}=0{,}2269828$	10
Варіант 5		***Варіант 13***	
$I=\int\limits_3^5 x\cdot e^x dx$, $I_{\text{іст.}}=553{,}4815626$	10	$I=\int\limits_2^7\frac{\ln x}{(x+1)^2}dx$, $I_{\text{іст.}}=0{,}259744$	10
Інтеграл та його істинне значення	***n***	**Інтеграл та його істинне значення**	***n***
Варіант 6		***Варіант 14***	
$I=\int\limits_0^5(e^{-x}+\frac{1}{x+1})dx$, $I_{\text{іст.}}=2{,}7850215$	10	$I=\int\limits_5^{10}\frac{(x+2)^2}{e^x}dx$, $I_{\text{іст.}}=0{,}4302486$	10
Варіант 7		***Варіант 15***	
$I=\int\limits_1^2 e^{\cos x}dx$, $I_{\text{іст.}}=1{,}1127801$	10	$I=\int\limits_0^5\frac{\ln(x+1)}{x^2+1}dx$, $I_{\text{іст.}}=0{,}9251183$	10
Варіант 8		***Варіант 16***	
$I=\int\limits_2^5\ln(x^2+1)dx$, $I_{\text{іст.}}=7{,}604111$	10	$I=\int\limits_0^3\frac{dx}{(x+1)\cdot\ln(x+5)}$, $I_{\text{іст.}}=0{,}7704097$	10

Контрольні запитання

1. Графічна інтерпретація визначеного інтегралу.
2. Правило прямокутників розрахунку визначеного інтегралу.
3. Правило трапецій розрахунку визначеного інтегралу.
4. Правило Ромберга розрахунку визначеного інтегралу.
5. Правило Сімпсона розрахунку визначеного інтегралу.

Список рекомендованої літератури

1. Косинський В. І., Швець О. Ф. Сучасні інформаційні технології. К. : Знання, 2011. 318 с.
2. Грицунов О. В. Інформаційні системи та технології : навчальний посібник. Харків : ХНАМГ, 2010. 222 с.
3. Тлумачний словник з інформатики / Г. Г. Півняк, Б. С. Бусигін, М. М. Дівізінюк та ін. Дніпропетровськ : Нац. гірн. ун-т, 2008. 599 с.